Proof of the Frobenius norm minimal point of Fast Matrix Multiplication in Strassen's orbit

> $restart : with(LinearAlgebra) : with(Student[VectorCalculus]) : with(PolynomialTools) :$

L represents the left linear pre-additions performed by the original Strassen's algorithm on left-hand side of A*B
K represents all the transformation of any L, R or P matrix within Strassen's orbit

> $L := Matrix(7, 4, [[1, 0, 0, 1], [0, 1, 0, -1], [-1, 0, 1, 0], [1, 1, 0, 0], [1, 0, 0, 0], [0, 0, 0, 1], [0, 0, 1, 1]]) :$
$\mathrm{W} := \left\langle \langle r|x\rangle, \left\langle 0 \middle| \frac{1}{r} \right\rangle \right\rangle : \mathrm{V} := \left\langle \langle s|y\rangle, \left\langle 0 \middle| \frac{1}{s} \right\rangle \right\rangle :$
$K := KroneckerProduct(W, V) :$ L, W, V, K;

$$\begin{bmatrix} 1 & 0 & 0 & 1 \\ 0 & 1 & 0 & -1 \\ -1 & 0 & 1 & 0 \\ 1 & 1 & 0 & 0 \\ 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 \\ 0 & 0 & 1 & 1 \end{bmatrix}, \begin{bmatrix} r & x \\ 0 & \frac{1}{r} \end{bmatrix}, \begin{bmatrix} s & y \\ 0 & \frac{1}{s} \end{bmatrix}, \begin{bmatrix} r\,s & r\,y & x\,s & x\,y \\ 0 & \frac{r}{s} & 0 & \frac{x}{s} \\ 0 & 0 & \frac{s}{r} & \frac{y}{r} \\ 0 & 0 & 0 & \frac{1}{r\,s} \end{bmatrix} \tag{1.1.1}$$

It is easier to study the square of the Frobenius norm, first its gradient

> Nlk $:= MatrixNorm(L \cdot K, Frobenius, conjugate=false);$
$E := Nlk^2 :$
$fx := diff(E, x) : fy := diff(E, y) : fr := diff(E, r) : fs := diff(E, s) :$
$gradE := [fx, fy, fr, fs] :$

$$Nlk := \left(4\,r^2 s^2 + 3\,r^2 y^2 + \frac{r^2}{s^2} + \left(r\,y + \frac{r}{s}\right)^2 + 3\,x^2 s^2 + \left(-x\,s + \frac{s}{r}\right)^2 + \frac{s^2}{r^2} + \left(x\,y + \frac{1}{r\,s}\right)^2 + \left(\frac{x}{s} - \frac{1}{r\,s}\right)^2 + \left(-x\,y + \frac{y}{r}\right)^2 + \left(x\,y + \frac{x}{s}\right)^2 + x^2 y^2 + \frac{1}{r^2 s^2} + \left(\frac{y}{r} + \frac{1}{r\,s}\right)^2\right)^{1/2} \tag{1.2.1}$$

We have found a real root of the gradient and now check that this point is indeed an extremal point

> $explminpoint := simplify\left(subs\left(\left\{ r = root[4]\left(\frac{3}{4}\right) \right\}, subs\left(\{s = r\}, subs\left(\left\{ y = -\frac{2 \cdot s^3}{3}, x = \frac{2 \cdot r^3}{3} \right\}, [r, s, x, y] \right) \right) \right) \right):$
$subminpoint := solve([r, s, x, y] - explminpoint);$
$map(simplify, subs(subminpoint, gradE));$

$$subminpoint := \left\{ r = \frac{3^{1/4}\sqrt{2}}{2}, s = \frac{3^{1/4}\sqrt{2}}{2}, x = \frac{3^{3/4}\sqrt{2}}{6}, y = -\frac{3^{3/4}\sqrt{2}}{6} \right\}$$

$$[0, 0, 0, 0] \tag{1.3.1}$$

We now compute the Hessian at that point (multiplying 9/(4) to simplify the following computations) , then the associated characteristic polynomial and eigenvalues

> $H := map\left(x \rightarrow simplify(radnormal(x, rationalized)), \frac{9}{4} \cdot Hessian(E, [r, s, x, y] = explminpoint) \right);$
$charP := FromCoefficientVector\big(map(x \rightarrow simplify(x, symbolic), CoefficientVector(simplify(expand(CharacteristicPolynomial(H, X)), radical), X)), X\big);$
$eigs := map(x \rightarrow simplify(x, symbolic), [solve(charP)]);$

$$H := \begin{bmatrix} 65\sqrt{3} & 23\sqrt{3} & 15 & -15 \\ 23\sqrt{3} & 65\sqrt{3} & 15 & -15 \\ 15 & 15 & 15\sqrt{3} & 3\sqrt{3} \\ -15 & -15 & 3\sqrt{3} & 15\sqrt{3} \end{bmatrix}$$

$$charP := X^4 - 160\sqrt{3}\,X^3 + 22536\,X^2 - 362880\sqrt{3}\,X + 5143824$$

$$eigs := \left[50\sqrt{3} + 4\sqrt{3}\sqrt{109}, 50\sqrt{3} - 4\sqrt{3}\sqrt{109}, 42\sqrt{3}, 18\sqrt{3}\right] \tag{1.4.1}$$

We end by checking that these eigenvalues are all positive, hence the Hessian is definite positive and the extremal point is a local minimum

> $map(x \rightarrow x > 0, eigs); evalf(\%); map(evalb, \%);$

$$\left[0 < 50\sqrt{3} + 4\sqrt{3}\sqrt{109}, 0 < 50\sqrt{3} - 4\sqrt{3}\sqrt{109}, 0 < 42\sqrt{3}, 0 < 18\sqrt{3}\right]$$

$$[0. < 158.9351057, 0. < 14.26997508, 0. < 72.74613394, 0. < 31.17691454]$$

$$[true, true, true, true] \tag{1.5.1}$$